

Berechnungsprogramme für Juristen mit Microsoft Windows 3.0 (Teil I)

Volker Nilgens

Mit der Ausgabe 2/92 haben die Abonnenten von jur-pc eine Diskette mit unterschiedlichen Berechnungsprogrammen für Windows 3.0 erhalten.

Ursprünglich dienten diese Programme dem Autor dazu, im Rahmen eines Beta-Testes für die Firma Borland GmbH das Programm Turbo Pascal für Windows in einem realistischen Umfeld zu testen. Die Programme wurden seither mehrfach weiterentwickelt und verbessert. Die jeweils aktuelle Version wird in der jur-pc Mailbox zum Download bereitgehalten.

Der Verfasser wird in loser Folge über die Besonderheiten der Programmierung für und unter Windows, die einzelnen Applikationen, ihre Rechenregeln und Bedienungsmöglichkeiten berichten. Die einzelnen Programme erheben nicht den Anspruch, die (einzig) richtige Berechnung zu sein; sie sollen jedoch als Grundlage für eine Diskussion der Rechenregeln und deren Umsetzung dienen.

Als erstes soll die Zinsberechnung im Mittelpunkt der Erläuterung stehen. Das Programm 'ZINS.EXE' ermöglicht die Berechnung von Anfangskapital, Endkapital oder Zinssatz. Es wird jeweils das nicht ausgefüllte Feld berechnet. Zur Vermeidung von Fehlermeldungen ist darauf zu achten, daß eines der vorgenannten Felder ohne Eintrag bleibt und die übrigen Felder ausgefüllt sind. Als Dezimaltrennzeichen sind '.' oder ',' erlaubt.

Die Berechnung des Zinszeitraums kann bankmäßig mit 360 Tagen pro Jahr und 30 Tagen pro Monat oder kalendermäßig erfolgen. Bei bankmäßiger Berechnung sind die Felder Jahre, Monate und Tage zu verwenden; bei kalendermäßiger Berechnung sind die Felder Zinsbeginn und Zinsende auszufüllen. Die Vorgaben bestimmen insofern die Art der Berechnung. Bei kalendermäßiger Berechnung kann die standardmäßige Einbeziehung von erstem und letztem Tag ausgeschaltet werden.

Zur Berechnung werden die nachfolgenden Abkürzungen und Funktionen verwendet:

i = Zinssatz (nomineller Jahreszins)
 K0 = Anfangskapital
 Kn = Endkapital
 n = Laufzeit (in Jahren)

E_* = Einfache Zinsrechnung
 Z_* = Zinseszinsrechnung
 G_* = Gemischte Zinsrechnung

Der '*' stellt einen Platzhalter für Zinssatz, Anfangs- oder Endkapital dar.

(* ***** allg. Prozeduren und Funktionen ***** *)

```
function hoch(basis, exponent: real): real;
begin
    hoch := exp(exponent * ln(basis));
end;
```

(* ***** Berechnung des Endkapitals (Kn) ***** *)

```
function E_Kn(K0, n, i: real): real;
begin
    E_Kn := K0 * (1 + n * i/100);
end;
```

```
function Z_Kn(K0, n, i: real): real;
begin
    Z_Kn := K0 * hoch((1 + i/100), n);
end;
```

*Erwünscht:
 Diskussionsbeiträge der Leser*

*Berechnung von
 Anfangs-/Endkapital oder
 Zinssatz*



*Zinszeitraum:
 Bank- oder kalendermäßige
 Berechnung*

Abkürzungen und Funktionen

Prozeduren und Funktionen

Berechnung des Endkapitals

Volker Nilgens arbeitet als wissenschaftlicher Mitarbeiter am rechtswissenschaftlichen Seminar der Universität Köln und studiert Informatik. Einigen Lesern ist er durch seine Programme für Juristen bekannt, die auch in der jur-pc Mailbox zum Download bereitliegen.



```
function G_Kn(K0, n, i: real): real;
var
  n1,
  n2 : real;
begin
  n1 := int(n);           { Ermittlung der ganzen Jahre }
  n2 := n - n1;          { Bestimmung der restlichen Zeit }
  G_Kn := E_Kn(Z_Kn(K0, n1, i), n2, i);
end;
```

(* ***** Berechnung des Anfangskapitals (K0) ***** *)

*Berechnung des
Anfangskapitals*

```
function E_K0(Kn, n, i: real): real;
begin
  E_K0 := Kn / (1 + n * i/100);
end;
```

```
function Z_K0(Kn, n, i: real): real;
begin
  Z_K0 := Kn * hoch((1 + i/100), -n);
end;
```

```
function G_K0(Kn, n, i: real): real;
var
  n1,
  n2 : real;
begin
  n1 := int(n);           { Ermittlung der ganzen Jahre }
  n2 := n - n1;          { Bestimmung der restlichen Zeit }
  G_K0 := E_K0(Z_K0(Kn, n1, i), n2, i);
end;
```

(* ***** Berechnung des Zinssatzes (i) ***** *)

Berechnung des Zinssatzes

```
function E_i(K0, Kn, n: real): real;
begin
  E_i := ((1 / n) * (Kn / K0 - 1)) * 100;
end;
```

```
function Z_i(K0, Kn, n: real): real;
begin
  Z_i := (hoch((Kn / K0), (1 / n)) - 1) * 100;
end;
```

```
function G_i(K0, Kn, n: real): real;
var
  z1 : real;
begin
  z1 := Z_i(K0, Kn, n);
  while G_Kn(K0, n, z1) < Kn do
    z1 := z1 - 0.001;    { Annäherung an den tatsächlichen Wert }
  G_i := z1;
end;
```